


9/8.5.1

Programme de Dump en Basic

Analysons la structure du programme de Dump Basic.

- 1) Passage en MODE 1.
- 2) Entrée des informations concernant le Dump :
 - adresse de début de Dump (en décimal ou hexa. [précédé de &]) ;
 - adresse de fin de Dump (en décimal ou hexa. [précédé de &]) ;
 - choix de la sortie (écran ou imprimante).
- 3) Affichage/impression des données lues en mémoire :
 - adresse du premier octet ;
 - 16 octets consécutifs ;
 - valeurs ASCII de ces 16 octets ;
 - test de fin d'affichage/impression ;
 - poursuite éventuelle de l'affichage/impression.

Ce qui se traduit par l'organigramme suivant :



Le listing du programme est le suivant :

```
1000 REM -----
1010 REM -           DUMP Hexadecimal et ASCII           -
1020 REM -----
1030 '
1040 '=====
1050 ' Entree des donnees
1060 '=====
1070 MODE 1
1080 PRINT "Utilitaire de DUMP"
1090 LOCATE 1,4:INPUT"A partir de : ";AD
1100 IF AD>2^16-1 THEN 1090
1110 LOCATE 1,6:INPUT"jusqu'a : ";AF
1120 IF AF>2^16-1 THEN 1110
1130 LOCATE 1,8: PRINT "1) Affichage sur ecran,"
1140 PRINT "2) impression sur imprimante."
1150 INPUT "Votre choix : ";CH
1160 IF CH<>1 AND CH<>2 THEN 1130
1170 IF CH=1 THEN PER=1 ELSE PER=8
1180 '
2000 '=====
2010 ' Affichage des donnees lues
2020 '=====
2030 MODE 2
2040 IF PER=8 THEN LOCATE 25,13:PRINT"Impression en cours..."
2050 FOR I=AD TO AF STEP 16
2060   LIGNE=LIGNE+1 'Indicateur de ligne d'affichage
2070   AS$="" 'Initialisation de la ligne des codes ASCII
2080   IF I<2^4 THEN AD$="000"+HEX$(I):GOTO 2110
2090   IF I<2^8 THEN AD$="00"+HEX$(I):GOTO 2110
2100   IF I<2^12 THEN AD$="0"+HEX$(I) ELSE AD$=HEX$(I)
2110   PRINT #PER,AD$;" "; 'Affichage de la premiere adresse
```

```

2120   FOR J=0 TO 15
2130     A=PEEK(I+J)
2140     B#=CHR$(A)
2150     IF A<32 THEN B#="."
2160     IF (A>128 AND PER=8) THEN B#="."
2170     AS#=AS#+B#
2180     IF A<16 THEN A#="0"+HEX$(A) ELSE A#=HEX$(A)
2190     PRINT #PER,A#;" ";
2200   NEXT J
2210   PRINT #PER,"      ";AS#
2220   IF (LIGNE<>23 OR I>=AF) OR (PER=8) THEN 2260
2230   LOCATE 10,25:PRINT "Appuyez sur une touche pour voir la suite"
2240   a#=INKEY$:IF a#="" THEN 2240
2250   CLS #PER:LOCATE 1,1:LIGNE=0 'A nouveau 1ere ligne d'affichage
2260 NEXT I
2270 IF PER=1 THEN LOCATE 1,24 ELSE CLS

```

L'exécution du programme en demandant la sortie des données sur imprimante pour un Dump entre #0001 et #00C5 (par exemple) donne le résultat suivant :

```

0001  B9 7F ED 49 C3 7B 05 C3 8A B9 C3 84 B9 C5 C9 C3      ..I.é.....
0011  1D BA C3 17 BA D5 C9 C3 C7 B9 C3 B9 B9 E9 00 C3      .....
0021  C6 BA C3 C1 B9 00 00 C3 35 BA 00 ED 49 D9 FB C7      .....5...I...
0031  D9 21 2B 00 71 18 08 C3 41 B9 C9 00 00 00 00 CA      .!+.q...A.....
0041  00 00 00 00 00 00 9C 04 00 00 00 20 A9 20 19 19      .....
0051  2C 19 0D 01 BF 22 49 6D 70 72 65 73 73 69 6F 6E      ,...."Impression
0061  20 65 6E 20 63 6F 75 72 73 2E 2E 2E 22 00 00 00      en cours..."...
0071  00 00 00 D3 EF 03 00 00 41 D3 F4 FF 03 28 0D 00      .....A....(..
0081  00 C1 29 20 00 00 00 00 00 00 00 00 00 00 00      ..) .....
0091  00 00 00 00 00 00 00 00 00 00 00 00 00 00 00      .....
00A1  00 00 00 00 00 00 00 00 00 00 00 00 00 00 00      .....
00B1  00 00 00 00 00 00 00 00 00 00 00 00 00 00 00      .....
00C1  00 00 00 00 00 00 00 00 00 00 00 00 00 00 00      .....

```